

Nr. 2

Excel tidsskriftet

Oktober 2022

Bo Jönsson - webbojo
17-10-2022

Indledning

Anden udgave af Excel tidsskriftet indeholder forskellige spændende funktioner og projekter.

Funktioner i kategorien Tekst er ikke dem jeg besøger oftest, men jeg har fundet et par stykker frem denne gang: STORT.FORBOGSTAV og MIDT. Der findes en del i denne kategori og kan med fordel undersøges nærmere.

Et interessant og noget kompliceret projekt kan du finde i afsnittet Dynamisk kalenderformel. Her vil du se hvordan man kan oprette en kalender, der opdateres – altså er dynamisk! Her introduceres du for flere forskellige funktioner.

Det ovenstående samt meget andet kan du blive klogere på i denne udgave af Excel tidsskriftet!

God fornøjelse

Bo

Indhold

Lad det første bogstav være stort i en tekst med Excel	3
Generisk formel	3
Resumé	3
Forklaring	3
Tildel data tilfældigt til grupper	4
Generisk formel	4
Resumé	4
Forklaring	4
Automatisk genberegning	4
Dynamisk kalenderformel i Excel	5
Resumé	5
Forklaring	5
Excel ENTYDIGE funktion	11
Resumé	11
Formål	11
Returværdi	11
Syntaks	11
Argumenter	11
Noter	12
Præcis én gang	12
Tæl kolonner, der indeholder specifikke værdier i Excel	14
Generisk formel	14
Resumé	14
Forklaring	14
BYCOL	15

Lad det første bogstav være stort i en tekst med Excel

Lad det første bogstav være stort i en tekst med Excel

	A	B	C	D
1	Tekst før		Tekst efter	
2	video er en kraftfuld metode til at understrege		Video er en kraftfuld metode til at understrege	
3	når du klikker på onlinevideo, kan du indsætte i		Når du klikker på onlinevideo, kan du indsætte integre	
4	du kan også skrive et nøgleord, hvis du vil søge o		Du kan også skrive et nøgleord, hvis du vil søge online o	
5	du kan f.eks. tilføje matchende forside, overskri		Du kan f.eks. tilføje matchende forside, overskrift eller	
6	klik på indsæt for at vælge mellem elementerne		Klik på indsæt for at vælge mellem elementerne i de fo	
7	temaer og typografier hjælper også med til at h		Temaer og typografier hjælper også med til at holde dc	
8	når du klikker på design og vælger et nyt tema,		Når du klikker på design og vælger et nyt tema, ændres	
9	når du anvender typografier, ændres dine overs		Når du anvender typografier, ændres dine overskrifter	
10	spar tid i word med nye knapper, der vises, når		Spar tid i word med nye knapper, der vises, når du har	
11				
12				

Generisk formel

=STORT.FORBOGSTAV(VENSTRE(A1))&MIDT(A1;2;LÆNGDE(A1))

Resumé

For at skrive det første bogstav i et ord eller en streng med stort, kan du bruge en formel baseret på funktionerne VENSTRE, MIDT og LÆNGDE. I det viste eksempel er formelen i C2:

=STORT.FORBOGSTAV(VENSTRE(A2))&MIDT(A2;2;LÆNGDE(A2))

Forklaring

Det første udtryk bruger VENSTRE og STORT.FORBOGSTAV til at skrive det første bogstav med stort:

=STORT.FORBOGSTAV(VENSTRE(A2))

Det er ikke nødvendigt at indtaste 1 for antal_tegn i VENSTRE, da det som standard vil være 1. Det andet udtryk udtrækker de resterende tegn med MID:

MIDT(A1;2;LÆNGDE(A1))

Teksten kommer fra A2, starttallet er hardkodet som 2, og antal_tegn leveres af funktionen LÆNGDE. Teknisk set behøver vi kun at udtrække (længde - 1) tegn, men MIDT vil ikke klage, hvis vi beder om flere tegn, så vi har efterladt tingene i den enkleste form.

Små bogstaver i resten

Hvis du vil have alt med små bogstaver undtagen det første bogstav, skal du blot indsætte det andet udtryk i funktionen SMÅ.BOGSTAVER:

=STORT.FORBOGSTAV(VENSTRE(A1))&SMÅ.BOGSTAVER(MIDT(A1;2;LÆNGDE(A1)))

Funktionen SMÅ.BOGSTAVER tvinger de resterende tegn til små bogstaver.

Tildel data tilfældigt til grupper

	A	B	C	D	E
1	Nummer	Navn	Gruppe		
2		1 Alfa	Tredje		
3		2 Beta	Anden		
4		3 Gamma	Første		
5		4 Epsilon	Anden		
6					

Generisk formel

=VÆLG(SLUMPMELEEM(1;3); "Gruppe A"; "Gruppe B"; "Gruppe B")

Resumé

For tilfældigt at finde personer (eller tilsvarende) i grupper kan du bruge funktionen SLUMPMELEEM med funktionen VÆLG. I det viste eksempel er formlen i C2:

=VÆLG(SLUMPMELEEM(1;3); "Første"; "Anden"; "Tredje")

Når den er kopieret ned i kolonnen, vil denne formel generere en tilfældig gruppe (Første, Anden eller Tredje for hvert nummer på listen).

Forklaring

Funktionen SLUMPMELEEM genererer tilfældige tal mellem to heltal. Funktionen VÆLG tager et tal som det første argument og bruger det tal til at vælge det "n'te" element fra de efterfølgende argumenter.

Så i denne formel genererer SLUMPMELEEM et tal mellem 1 og 3, dette tal bruges til at vælge en gruppe fra de 3 følgende værdier: "Første"; "Anden"; "Tredje".

Du kan bruge den samme tilgang, når som helst du skal lave tilfældige opgaver. Det er især nyttigt, når du skal tildele data til et begrænset antal tekstværdier.

Automatisk genberegning

Vær opmærksom på, at SLUMPMELEEM vil genberegnes, når der foretages ændring i en projektmappe, eller endda når en projektmappe åbnes. Når du har et sæt tilfældige tildelinger, vil du måske kopiere og indsætte formlerne som værdier for at forhindre yderligere ændringer.

Dynamisk kalenderformel i Excel

	A	B	C	D	E	F	G	H	I	J	K	L	M
1		januar 2022								Start		Helligdage	
2		26	27	28	29	30	31	1		01-01-2022		01-01-2022	Nytårsdag
3		2	3	4	5	6	7	8				06-01-2022	Hellig tre konger
4		9	10	11	12	13	14	15				14-02-2022	Valentinsdag
5		16	17	18	19	20	21	22				27-02-2022	Fastelavn
6		23	24	25	26	27	28	29				27-03-2022	Sommertid
7		30	31	1	2	3	4	5				10-04-2022	Palmesøndag
8												14-04-2022	Skærtorsdag
9												15-04-2022	Langfredag
10												17-04-2022	Påske
11												17-04-2022	Påskedag
12												18-04-2022	2. Påskedag
13												08-05-2022	Mors dag
14												13-05-2022	Store Bededag

Resumé

For at oprette en dynamisk månedskalender ved hjælp af en enkelt formel, kan du bruge funktionen SEKVENS, med hjælp fra funktionerne VÆLG og UGEDAG. I det viste eksempel er formelen i B6:

```
=SEKVENS(6;7;Start-VÆLG(UGEDAG(Start);0;1;2;3;4;5;6))
```

hvor Start er det navngivne område J2. I det viste eksempel bruges betinget formatering til at fremhæve den aktuelle dato og helligdage og lysere dage i andre måneder. Se nedenfor for en fuldstændig forklaring.

Bemærk: Dynamiske array-funktioner er kun tilgængelige i Excel 365 og 2021.

Forklaring

Bemærk: Dette eksempel antager, at startdatoen vil blive angivet som den første i måneden. Se nedenfor for en formel, der automatisk returnerer den første dag i den aktuelle måned.

I dette eksempel er målet at generere en dynamisk kalender for en given måned, baseret på en startdato indtastet i celle J2, som er navngivet "Start" Vi antager, at Start er en gyldig første-i-måned-dato som 01-01-22, 01-02-22, 01-03-22, 01-04-22 osv. Den endelige kalender skal placere hver dag i måneden i et skema, hvor hver uge starter på søndag, som det ses i eksemplet. Løsningen forklaret nedenfor er baseret på funktionen SEKVENS. SEKVENS er en af de originale dynamiske array-funktioner i Excel og passer perfekt til denne opgave.

Kort version

Forklaringen nedenfor er ret lang. Den korte version er, at SEKVENS funktionen genererer en 6 x 7 matrix med 42 datoer i et kalendergitter, formateret til kun at vise dagen. Dette virker, fordi Excel-datoer kun er serietal. Den største udfordring med dette problem er at finde ud af, hvilken dato man skal starte med for en given måned, som altid er en søndag. Dette håndteres med funktionerne VÆLG og UGEDAG. Betinget formatering bruges til at fremhæve den aktuelle dato og helligdage og lysere dage i andre måneder.

Grundsekvens

Funktionen SEKVENS kan bruges til at generere numeriske sekvenser. For eksempel, for at generere tallene 1 til 10 i ti rækker, kan du bruge SEKVENS sådan her:

```
=SEKVENS(10) // returnerer {1;2;3;4;5;6;7;8;9;10}
```

Excel ENTYDIGE funktion

Resultatet er en matrix, der indeholder tallene 1-10. Arrayet vises i et lodret område på ti celler. SEKVENS kan generere arrays i rækker og kolonner. For eksempel opretter følgende formel med tallene 1-10 i en matrix med 5 rækker og 2 kolonner:

```
=SEKVENS (5 ; 2)
```

Og formlen nedenfor vil fylde et 7 x 6 gitter af celler med tallene 1-42:

```
=SEKVENS (6 ; 7)
```

Skærbilledet nedenfor viser, hvordan disse formler opfører sig i et regneark:

	A	B	C	D	E	F	G	H	I	J	K	L
1	=SEKVENS(10)		=SEKVENS(5;2)			=SEKVENS(6;7)						
2												
3	1		1	2		1	2	3	4	5	6	7
4	2		3	4		8	9	10	11	12	13	14
5	3		5	6		15	16	17	18	19	20	21
6	4		7	8		22	23	24	25	26	27	28
7	5		9	10		29	30	31	32	33	34	35
8	6					36	37	38	39	40	41	42
9	7											
10	8											
11	9											
12	10											
13												

Det er kun tal, ikke datoer, men du kan se kernekonceptet.

SEKVENS med datoer

Fordi Excel-datoer kun er store serienumre, kan funktionen SEKVENS nemt bruges til at generere arrays af datoer. For eksempel vil formlen nedenfor generere de første datoer for de 31 dage i januar 2023:

```
=SEKVENS (31 ; 1 ; DATO (2022 ; 1 ; 1) )
```

Bemærk: funktionen DATO er en mere sikker måde at fastkode datoer i formler, da datoer indtastet som tekst kan misfortolkes.

For at oversætte: vi beder om 31 tal, i et 31 x 1 array, startende med 1. januar 2022. SEKVENS indstiller automatisk til en trinværdi på 1, så resultatet er en liste over serietal, der starter med 44265. Det er klart, at vi ønsker ikke at vise serietal i vores kalender, vi ønsker at vise dage. For at gøre det kan vi bruge det brugerdefinerede talformat "d". Det vil få Excel til kun at vise dagtallene. Skærbilledet nedenfor viser før og efter:

Excel ENTYDIGE funktion

	A	B	C	D	E	F	G
1	Uden formatering				Med talformatering "d"		
2	=SEKVEN(31;1;DATO(2022;1;1))				=SEKVEN(31;1;DATO(2022;1;1))		
3	44562				1		
4	44563				2		
5	44564				3		
6	44565				4		
7	44566				5		
8	44567				6		
9	44568				7		
10	44569				8		
11	44570				9		
12	44571				10		

Lad os nu se, hvad der sker, hvis vi beder om 6 x 7 gitter, startende med 1. januar 2022:

=SEKVEN(6;7;DATO(2022;1;1))

Når vi først har formateret outputtet med det brugerdefinerede talformat "d", ser vi i alt 42 tal, begyndende med 1. januar. I slutningen af januar ændres måneden til februar, og dagen bliver igen 1:

	A	B	C	D	E	F	G	H
1		=SEKVEN(6;7;DATO(2022;1;1))						
2		1	2	3	4	5	6	7
3		8	9	10	11	12	13	14
4		15	16	17	18	19	20	21
5		22	23	24	25	26	27	28
6		29	30	31	1	2	3	4
7		5	6	7	8	9	10	11
8								
9		Cellerne er formateret med det brugerdefinerede talformat "d"						
10								

Vi har stadig ikke en brugbar kalender, men vi nærmer os!

For at lave en ordentlig kalender, skal den første dag i vores skema starte på søndag. Hvis den første dag i en måned ikke er en søndag, skal vi starte skemaet den sidste søndag i den foregående måned. Hvordan kan vi beregne den sidste søndag i den foregående måned? Før vi kommer ind på specifikke funktioner, lad os præcisere målet.

Første søndag

Hvis den første i en måned tilfældigvis er en søndag, er vi færdige. Der er ingen grund til at gøre noget. Den første i måneden er vores startdato. Men hvis den første i måneden ikke er en søndag, skal vi "rulle tilbage" et antal dage til den foregående søndag. Hvor mange dage skal vi rulle tilbage? Dette afhænger af, hvilken ugedag den første dag i en måned lander på. For eksempel, hvis den første er en tirsdag, skal vi rulle 2 dage

Excel ENTYDIGE funktion

tilbage. Hvis den første er en fredag, skal vi rulle 5 dage tilbage. Og hvis den første allerede er en søndag, skal vi rulle 0 dage tilbage.

Nu har vi en ret god idé om, hvad vi skal gøre, vi skal bare implementere den adfærd i en formel. Det er her, formlen bliver lidt tricky, fordi vi skal kombinere to funktioner, UGEDAG og VÆLG, på en måde, som de fleste brugere ikke vil genkende.

Funktionen UGEDAG

For at finde ud af ugedagen bruger vi funktionen UGEDAG. UGEDAG returnerer et tal for hver dag i ugen. Som standard returnerer UGEDAG 1 for søndag og 7 for lørdag. For eksempel returnerer UGEDAG 1 for 1. januar 2023, da den første er en søndag:

```
=UGEDAG(DATO(2023;1;1)) // returnerer 1
```

For den 2. januar 2022 returnerer UGEDAG 2, da den anden er en mandag:

```
=UGEDAG(DATO(2023;1;2)) // returnerer 2
```

For at opsummere vil UGEDAG give os et tal mellem 1-7 for hver dag i ugen, og vi kan bruge resultatet til at finde ud af, hvor mange dage vi skal rulle tilbage.

Funktionen VÆLG

Funktionen VÆLG bruges til at vælge vilkårlige værdier efter numerisk position. For eksempel, hvis vi har farverne "rød", "blå" og "grøn", kan vi bruge VÆLG på denne måde:

```
=VÆLG(1;"rød";"blå";"grøn") // returnerer "rød"
```

```
=VÆLG(2;"rød";"blå";"grøn") // returnerer "blå"
```

```
=VÆLG(3;"rød";"blå";"grøn") // returnerer "grøn"
```

VÆLG er en fleksibel funktion og accepterer en liste af tekstværdier, tal, cellereferencer i enhver kombination.

VÆLG + UGEDAG

Dernæst vil vi kombinere VÆLG og UGEDAG for at give os det korrekte tal der skal ruller tilbage, som dette:

```
=VÆLG(UGEDAG(Start);0;1;2;3;4;5;6)
```

Argumentet *serienr* leveres af funktionen UGEDAG. De andre individuelle værdier, der gives til VÆLG, er tallene der skal ruller tilbage, en for hver dag i ugen. UGEDAG returnerer et tal mellem 1-7, og funktionen VÆLG bruger tallet fra UGEDAG til at vælge et tal fra listen over tal. For eksempel, hvis UGEDAG returnerer 3 (tirsdag), returnerer VÆLG 2:

```
VÆLG(3;0;1;2;3;4;5;6) // returnerer 2
```

Nu er vi endelig klar til at bruge dette tilbagerulningstal til at beregne den første søndag i skemaet. Fordi datoer kun er tal i Excel, er opgaven enkel - vi skal bare trække tilbagerulningstallet fra startdatoen:

```
Start-VÆLG(UGEDAG(Start);0;1;2;3;4;5;6) // første søndag
```

Resultatet er en gyldig dato, der repræsenterer den første søndag i kalenderskemaet.

Excel ENTYDIGE funktion

Vi samler det hele

Nu skal vi kombinere ideerne forklaret ovenfor i en enkelt formel baseret på SEQUENCE-funktionen. Vi starter med at bede om et 6 x 7 array af tal som dette:

```
=SEKVEN(6,7 // 6 rækker, 7 kolonner
```

Så, til startargumentet, leverer vi blot den kode, vi udarbejdede ovenfor:

```
=SEKVEN(6;7;Start-VÆLG(UGEDAG(Start);0;1;2;3;4;5;6))
```

Resultatet er et komplet gitter med 42 datoer, der kan vises som en månedlig kalender. Hvis startdatoen i J2 ændres til en anden første i månedsdato, opdateres skemaet automatisk.

Regler for betinget formatering

De betingede formateringsregler for at fremhæve den aktuelle dato og helligdage og lysere dage i andre måneder er angivet nedenfor:

Formlen for den aktuelle dato er:

```
=B2=IDAG()
```

Formlen til at fremhæve helligdage er baseret på funktionen COUNTIF:

```
=TÆL.HVIS(Helligdage;B2)
```

Hvis tallet er andet end nul, skal datoen være en helligdag. Helligdage skal være et interval, der indeholder gyldige Excel-datoer, der repræsenterer ikke-arbejdsdage. I det viste eksempel er helligdage det navngivne område L6:L8. Du kan tilføje flere helligdage til denne liste, som du vil, men glem ikke at opdatere det navngivne område. Alternativt kan du definere helligdage som en Excel-tabel, så intervallet opdateres automatisk.

Formlen til dage i andre måneder er baseret på funktionen MÅNED:

```
=MÅNED(B6)<>MÅNED(Start)
```

Hvis måneden for den aktuelle dato er forskellig fra måneden for datoen i J2, skal du udløse reglen.

Kalender titel

Formlen til at udskrive kalendertitlen i celle B4 er baseret på TEKST-funktionen:

```
=TEKST(Start;"mmm åååå")
```

Titlen er centreret over kalenderskemaet med centreret på tværs af markeringen. Vælg B1:H1, og brug Ctrl + 1 til at åbne Formater celler, og vælg derefter "Centreret" fra rullemenuen Vandret tekstjustering. Dette er en bedre mulighed end at flette celler, da det ikke ændrer gitterstrukturen i regnearket.

Evighedskalender med aktuel dato

For at oprette en evighedskalender, der opdateres automatisk baseret på den aktuelle dato, skal vi opdatere formelen for at generere den første dag i den aktuelle måned med det samme. Den første dag i den aktuelle måned kan beregnes med funktionen EOMONTH på denne måde:

```
=SLUT.PÅ.MÅNED(IDAG();-1)+1
```

Du kan bruge denne formel direkte i celle J2, og kalenderen forbliver altid opdateret. For en alt-i-en-formel kan vi tilføje LET-funktionen som denne:

Excel ENTYDIGE funktion

```
=LET(start; SLUT.PÅ.MÅNED(IDAG();-1)+1;SEKVEN(6;7;Start-  
VÆLG(UGEDAG(Start);0;1;2;3;4;5;6))
```

Her bruger vi funktionen LET til at definere "Start" som den første dag i den aktuelle måned, hvorefter vi kører den oprindelige formel uændret. Den lokale variabel "Start" tilsidesætter den navngivne områdestart i regnearket, som kan slettes, hvis det ønskes.

Mandag

For at generere en kalender, der starter mandag i stedet for søndag, kan du bruge følgende kode inde i SEKVEN som startargument:

```
=Start-VÆLG(UGEDAG(Start);6;0;1;2;3;4;5)
```

Ved at bruge den samme logik forklaret ovenfor, ruller denne kode startdatoen tilbage efter behov for at begynde kalenderen på en mandag.

Excel ENTYDIGE funktion

	A	B	C
1	Værdier		Unikke værdier
2	16		16
3	18		18
4	15		15
5	20		20
6	17		17
7	18		12
8	18		11
9	12		10
10	11		19
11	10		
12	18		
13	16		
14	19		
15	11		
16	19		
17			

Resumé

Excel funktionen ENTYDIGE returnerer en liste over unikke værdier i en liste eller et område. Værdier kan være tekst, tal, datoer, tidspunkter osv.

Formål

Udtræk unikke værdier fra området

Returværdi

Array af unikke værdier

Syntaks

= ENTYDIGE(array, [efter_kol], [præcis_en_gang])

Argumenter

Array - Område eller array, hvorfra unikke værdier kan udtrækkes.

Efter_kol - [valgfrit] Hvordan der sammenlignes og uddrages. Efter række = FALSK (standard); efter kolonne = SAND.

Nøjagtig_en_gang - [valgfrit] TRUE = værdier, der forekommer én gang, FALSE = alle unikke værdier (standard).

Noter

Excel funktionen ENTYDIGE udtrækker en liste over unikke værdier fra et område eller et array. Resultatet er en dynamisk række af unikke værdier. Hvis dette array er det endelige resultat (dvs. ikke overføres til en anden funktion), vil array-værdier "overløbes" i regnearket til et område, der automatisk opdateres, når nye unikke værdier tilføjes eller fjernes fra kildeområdet.

Funktionen ENTYDIGE tager tre argumenter: Array, Efter_kol og Nøjagtig_en_gang. Det første argument, array, er arrayet eller området, hvorfra der skal udtrækkes unikke værdier. Dette er det eneste nødvendige argument. Det andet argument, Efter_kol, afgør, om ENTYDIGE skal udtrække unikke værdier efter rækker eller kolonner. Som standard vil ENTYDIGE udtrække unikke værdier i rækker. For at tvinge ENTYDIGE til at udtrække unikke værdier efter kolonner, skal du indstille Efter_kol til TRUE eller 1. Det sidste argument, Nøjagtig_en_gang, angiver adfærd for værdier, der vises mere end én gang. Som standard vil ENTYDIGE udtrække alle unikke værdier, uanset hvor mange gange de vises i arrayet. For at udtrække unikke værdier, der kun vises én gang i arrayet, skal du indstille Nøjagtig_en_gang til TRUE eller 1.

Bemærk: funktionen ENTYDIGE skelner ikke mellem store og små bogstaver. ENTYDIGE vil behandle "BANAN", "Banan" og "banan" som den samme tekst.

Simple eksempel

Funktionen ENTYDIGE udtrækker unikke værdier fra et område eller et array:

```
=ENTYDIGE ({"A";"B";"C";"A";"B"}) // returnerer {"A";"B";"C"}
```

For at returnere unikke værdier fra i området A1:A10, kan du bruge en formel som denne:

```
=ENTYDIGE (A1:A10)
```

Efter kolonne

Som standard vil ENTYDIGE udtrække unikke værdier i rækker:

```
=ENTYDIGE ({1;1;2;2;3}) // returnerer {1;2;3}
```

ENTYDIGE vil ikke håndtere de samme værdier organiseret i kolonner:

```
=ENTYDIGE ({1\1\2\2\3}) // returnerer {1,1,2,2,3}
```

For at håndtere det vandrette array ovenfor skal du indstille sættes Efter_kol argumentet til SAND eller 1:

```
=ENTYDIGE ({1\1\2\2\3};1) // returnerer {1,2,3}
```

For at returnere unikke værdier fra det vandrette område A1:E1 skal du indstille argumentet Efter_kol til SAND eller 1:

```
=ENTYDIGE (A1:E1;1) // ekstraher unik fra vandret array
```

Præcis én gang

Funktionen ENTYDIGE har et valgfrit argument kaldet præcis_en_gang, der styrer, hvordan funktionen håndterer gentagne værdier. Som standard er nøjagtig_en_gang FALSK. Dette betyder, at ENTYDIGE vil udtrække unikke værdier, uanset hvor mange gange de vises i kildedataene:

```
=ENTYDIGE ({1;1;2;2;3}) // returnerer {1;2;3}
```

Indstil nøjagtig_en_gang til TRUE eller 1 for at udtrække unikke værdier, der kun vises én gang i kildedataene:

Excel ENTYDIGE funktion

```
=ENTYDIGE({1;1;2;2;3};0;1) // returnerer {3}
```

Bemærk, at ovenstående formel også sætter argumentet Efter_kol til nul (0), standardværdien. Den samme formel kunne også skrives sådan:

```
=ENTYDIGE({1;1;2;2;3};;1)
```

```
=ENTYDIGE({1;1;2;2;3};0;SAND) // returnerer {3}
```

```
=ENTYDIGE({1;1;2;2;3};FALSK;SAND) // returnerer {3}
```

Unikt med kriterier

For at udtrække unikke værdier, der opfylder specifikke kriterier, kan du bruge ENTYDIGE sammen med FILTRER-funktionen. Den generiske formel, hvor omdr2=A1 repræsenterer en logisk test, ser sådan ud:

```
=ENTYDIGE(FILTRER(omrd1;omrd2=A1))
```

Tæl kolonner, der indeholder specifikke værdier i Excel

	A	B	C	D	E	F	G	H	I	J
1										
2		20	30	12	23	12		Værdi	20	
3		13	15	30	14	22				
4		14	11	13	21	23		Antal	3	
5		22	10	30	10	12				
6		12	21	21	26	30				
7		15	20	11	21	16				
8		20	25	13	18	29				
9		22	15	29	22	14				
10		26	14	28	20	22				
11		27	18	28	25	19				
12		18	14	24	22	24				
13		13	11	30	27	18				
14		29	26	26	26	11				
15		29	20	28	25	18				
16		29	21	23	21	13				
17										
18										

Generisk formel

`=SUM(--(MPRODUKT(TRANSPONER(RÆKKE(data)^0);--(kriterie))>0))`

Resumé

For at tælle rækker, der indeholder specifikke værdier, kan du bruge en formel baseret på funktionerne MPRODUKT, TRANSPONER, RÆKKE og SUM. I det viste eksempel er formlen i I4:

`=SUM(--(MPRODUKT(TRANSPONER(RÆKKE(Data)^0);--(Data=I2))>0))`

hvor Data er det navngivne område B2:F16. Resultatet er 3, antallet af kolonner, der indeholder tallet 20.

Forklaring

I dette eksempel er målet at tælle antallet af kolonner i dataene, der indeholder 20 (værdien i celle I2). Den største udfordring i dette problem er, at værdien kan forekomme i en hvilken som helst række og mere end én gang i samme kolonne. Hvis vi blot ville tælle det samlede antal gange, en værdi optrådte i et interval, kunne vi bruge funktionen TÆL.HVIS. Men vi har brug for en mere avanceret formel til at tælle kolonner, der kan indeholde flere forekomster af en bestemt værdi. Forklaringen nedenfor gennemgår to muligheder: en baseret på funktionen MPRODUKT og en baseret på den nyere funktion BYCOL.

MPRODUKT

En mulighed for at løse dette problem er funktionen MPRODUKT. Funktionen MPRODUKT returnerer matrixproduktet af to arrays, nogle gange kaldet "Skalarproduktet". Resultatet fra MPRODUKT er et array, der indeholder det samme antal rækker som array1 og det samme antal kolonner som array2. Funktionen MPRODUKT tager to argumenter, matrix1 og matrix2, som begge er påkrævet. Kolonneantallet i matrix1 skal svare til rækkeantallet i matrix2. I det viste eksempel er formlen i I4:

`=SUM(--(MPRODUKT(TRANSPONER(RÆKKE(Data)^0);--(Data=I2))>0))`

Tæl kolonner, der indeholder specifikke værdier i Excel

	A	B	C	D	E	F	G	H	I
1									
2		20	30	12	23	12		Værdi	20
3		13	15	30	14	22			
4		14	11	13	21	23		Antal	3
5		22	10	30	10	12			
6		12	21	21	26	30			
7		15	20	11	21	16			
8		20	25	13	18	29			
9		22	15	29	22	14			
10		26	14	28	20	22			
11		27	18	28	25	19			
12		18	14	24	22	24			
13		13	11	30	27	18			
14		29	26	26	26	11			
15		29	20	28	25	18			
16		29	21	23	21	13			
17									

Funktionen BYCOL gentager de navngivne områdedata (B2:F16) én kolonne ad gangen. Ved hver kolonne evaluerer og gemmer BYCOL resultatet af den medfølgende LAMBDA-funktion:

```
LAMBDA(kol;--(SUM(--(kol=I2))>0))
```

Ved at arbejde indefra og ud, kontrollerer logikken her for værdier i kol, der er lig med I2, hvilket resulterer i en matrix af SAND og FALSK værdier. SAND- og FALSK-værdierne tvinges til 1'er og 0'er med den dobbelte negativ (--), og SUM-funktionen summerer resultatet. Dernæst kontrollerer vi, om resultatet fra SUM er >0 og tvinger det resultat til 1 eller 0 med en anden dobbelt negativ. Efter at BYCOL er kørt, har vi et array med ét resultat pr. kolonne, enten 1 eller 0:

```
{1,1,0,1,0} // resultat fra BYCOL
```

Formlen kan nu forenkles som følger:

```
=SUM({1;1;0;1;0}) // returnerer 3
```

I det sidste trin summerer funktionen SUM elementerne i arrayet og returnerer et slutresultat på 3.

Tekst indhold

Hvis du har brug for at tjekke for specifikke tekstværdier, med andre ord bogstaveligt talt kontrollere, om celler indeholder bestemte tekstværdier, kan du ændre logikken i formelen på denne side for at bruge ER.TAL og SØG funktionen. For at tælle celler/rækker, der indeholder "banan", kan du f.eks. bruge:

```
=ER.TAL(SØG("banan";Data))
```